1 Développée d'une courbe

1.1 Enoncé

Il s'agit ici d'écrire une procédure `Develop` qui trace sur un même graphe une courbe donnée en paramétriques (de paramètre t) et sa développée, c'est à dire l'ensemble de ses centres de courbure.

- Ecrire des formules donnant directement les coordonnées u et v du centre de courbure en fonction de x, y et leurs dérivées.
- Ecrire une procédure `Develop` à 4 paramètres (x,y,a,b) où x et y sont des expressions de t et où $[a,b]$ est intervalle de variation de t.
  Cette procédure trace sur un même graphe la courbe et sa développée, dans un repère orthonormal.
  On pourra utiliser `diff` pour calculer les dérivées premières et secondes de x et y.
- Essayer cette procédure avec une ellipse : $x = 2\cos t$, $y = \sin t$ sur $[0,6.29]$.
- Essayer cette procédure pour une parabole : $x = t^2$, $y = t$ sur $[-0.5,0.5]$.

1.2 Corrigé

On établit d'abord des formules donnant les coordonnées du centre de courbure de la courbe en un point. Ceci figure d'ailleurs directement dans le cours :

$$u = x - y' \frac{x'^2 + y'^2}{\begin{vmatrix} x' & x'' \\ y' & y'' \end{vmatrix}}$$

$$v = y + x' \frac{x'^2 + y'^2}{\begin{vmatrix} x' & x'' \\ y' & y'' \end{vmatrix}}$$

```
Develop := proc(x, y, a, b)
local x1, x2, y1, y2, n2, d, u, v;
    x1 := diff(x, t);
    x2 := diff(x1, t);
    y1 := diff(y, t);
    y2 := diff(y1, t);
    n2 := x1^2 + y1^2;
    d := x1 * y2 - y1 * x2;
    u := x - n2 * y1/d;
    v := y + n2 * x1/d;
    plot({[x, y, t = a..b], [u, v, t = a..b]}, scaling = constrained)
end proc
```

Explication des variables :
- $x1$,$y1$,$x2$,$y2$ sont les dérivées premières et secondes de x et y,
- d est le déterminant des vecteurs dérivées premières et secondes,
- $n2$ est le carré de la norme du vecteur dérivée,
- u et v sont les coordonnées du point courant de la développée.

On n'oublie pas de se placer dans un repère orthonormal.

2 Podaire d'une courbe par rapport à un point

2.1 Enoncé

Il s'agit ici d'écrire une procédure `podaire` qui trace sur un même graphe une courbe donnée en paramétrique (de paramètre t) et sa podaire par rapport à un point.
La podaire d'une courbe par rapport à un de ses point est l'ensemble des projections orthogonales de ce point sur les tangentes à la courbe.

– Ecrire des formules donnant directement les coordonnées u et v de la projection orthogonale du point C de coordonnées (x_c,y_c) sur la tangente en M (de coordonnées (x,y)) en fonction de x, y, leurs dérivées, x_c et y_c.
– Ecrire une procédure `podaire` à 5 paramètres (x,y,a,b,c) où x et y sont des expressions de t, où $[a,b]$ est intervalle de variation de t et c la valeur du paramètre t en C.
Cette procédure trace sur un même graphe la courbe et sa podaire par rapport à C, dans un repère orthonormal.
On pourra utiliser `subs` pour calculer les coordonnées de C.
– Essayer cette procédure avec une ellipse : $x = 2\cos t$, $y = \sin t$ sur $[0,6.29]$ avec $c = 0$.
– Essayer cette procédure pour une parabole : $x = t^2$, $y = t$ sur $[-1,1]$ avec $c = 0.3$.

2.2 Corrigé

Le seul problème est d'établir les formules donnant directement la projection de C sur la tangente en un point quelconque.
On écrit d'abord l'équation de la tangente à la courbe en un point quelconque (x,y) :

$$-y'\,(X-x)+x'\,(Y-y)=0$$

On écrit ensuite l'équation de la normale à cette tangente passant par C :

$$x'\,(X-x_c)+y'\,(Y-y_c)=0$$

Le point cherché de la podaire est l'intersection de ces deux droites, on résout donc le système en X,Y pour obtenir les u et v de l'énoncé.

```
podaire := proc(x, y, a, b, c)
local x1, y1, xc, yc, u, v;
    xc := subs(t = c, x);
    yc := subs(t = c, y);
    x1 := diff(x, t);
    y1 := diff(y, t);
    u := (x*y1^2 - y*x1*y1 + xc*x1^2 + yc*x1*y1)/(x1^2 + y1^2);
    v := (y*x1^2 - x*y1*x1 + yc*y1^2 + xc*y1*x1)/(x1^2 + y1^2);
    plot({[x, y, t = a..b], [u, v, t = a..b]}, scaling = constrained)
end proc
```

Explication des variables :
– $x1$ et $y1$ sont les dérivées de x et y,
– xc et yc sont les coordonnées de C,
– u et v sont les coordonnées du point courant de la podaire.
On n'oublie pas de se placer en repère orthonormal.